Chapter 6
XML Query Languages

• XPath is not a query language:
selects only sets of nodes

• additional functionality of query languages:
– composition of tuples/structures from several nodes of a path
– joins
– dereferencing
  • via joins
  • via direct resolving of IDs (seen as values)
  • via dereferencing of ID attributes
– aggregations
– formatting and restructuring of results
– operations on the order of nodes!

Collected experiences from SQL, OQL, OEM/WSL/MSL, F-Logic and some more ...

• predecessors of XPath: XSL Patterns/XPointer/XQL (1998)
• XQL extended the early “basic form” to a query language
  – adding several constructs to the path expressions
  – increasingly complicated
  – still not sufficiently expressive
  – showed the limits and requirements
• XML-QL (1998): pattern-matching-based “extraction language”
  – not path-based, but XML-pattern/template-based binding of variables
  – semantics by a clause-construct
  – generation and structuring of the result by an XML pattern with variables
XML Query Languages (cont’d)

- Quilt (2000): SQL-style extension of XPath
  - binding of variables by XPath expressions
  - nested loops by “for”-clauses
  - additional conditions in a “where”-clause
  - structuring of the result by a “return”-clause

- XQuery (2001): “official” version of Quilt
  - W3C Working Draft XQuery first version from 15 February 2001
  - XQuery 1.0: W3C Recommendation since 23.1.2007
  - http://www.w3.org/TR/xquery/

6.1 XQL

XQL (XML Query Language; 1998) is a simple query language based on early constructs of XPath:

- all XPath expressions that can be expressed without the use of “axis::” (cf. Slide 199 - axes have only been added later).
- text() was a function,
- function applications have been expressed by “!” at the end of the path expression: 
  //country/name!text()

Further querying functionality was integrated syntactically into the path expressions.
XQL: B**OOLEAN O**P**ERATIONS AND S**ET O**P**ERATIONS**

- $q_1 \cup q_2$, $q_1 \lor q_2$
- $q_1 \cap q_2$
- $q_1 \sim q_2$ (union, in case that both are non-empty)
- $q_1 \lor q_2$
- $q_1 \land q_2$

---

XQL: **R**E**T**URN O**P**ERATORS (P**ROJECTIONS ON THE PATH**)

Operators that output the node that is addressed at the given position:

- `??`: the complete node is added to the output structure (including attributes and subelements)
- `?:` only the element “hull” is added to the output

- `country/city[@isCountryCap]/name`
  - `<name>Berlin</name>`
  - `<name>Rome</name>`
- `country?[city[@isCountryCap]/name`
  - `<country> <name>Berlin</name> </country>`
  - `<country> <name>Rome</name> </country>`
- `country?[@car_code?]/city[@isCountryCap]/name`
  - `<country car_code="D"> <name>Berlin</name> </country>`
  - `<country car_code="I"> <name>Rome</name> </country>`
- `country?[[@car_code?]/city?[[@isCountryCap]/name/text()`
XQL: GROUPING

• copy a part of the original document structure:
  \( \text{path}_1 \{ \text{path}_2 \} \)

• without grouping:
  \[
  \text{country?[@car_code?]}/\text{city?}/\text{name!text()}
  \]
  \[
  <\text{country car_code="D"}>
  <\text{city}>Berlin</\text{city}>
  </\text{country}>
  <\text{country car_code="D"}>
  <\text{city}>Hamburg</\text{city}>
  </\text{country}>
  <\text{country car_code="D"}>
  <\text{city}>Munich</\text{city}>
  </\text{country}>
  \]

• with grouping:
  \[
  \text{country?[@car_code?] } \{/\text{city?}/\text{name!text()}\}
  \]
  \[
  <\text{country car_code="D"}>
  <\text{city}>Berlin</\text{city}>
  <\text{city}>Hamburg</\text{city}>
  <\text{city}>Munich</\text{city}>
  </\text{country}>
  \]

XPATH: SEMIJOINS ARE POSSIBLE

• Semi-joins via subqueries in the condition:
  \[\pi[A](r \bowtie s) ,\ A \subset \text{attr}(r)\]

Query: name of the continent where Germany is located:

/mondial/continent[@id =
    /mondial/country[@car_code="D"]
    /encompassed/@continent]
  /name!text()

Problems

• full joins with join conditions not possible
• no restructuring/generation of answer structure
**XQL: JOINS**

Asymmetric full joins expressed by *correlating variables* and “alternative”-construct:
Filters may contain variable assignments of the form

\[
\text{[$var := expr$]}
\]

that are then used in another condition

\[
\text{[expr' = $var$]}
\]

//organization?[$s := @headq] {name?? | abbrev?? | member?? | //city[@id=$s]?? }

<organization>
  <name>European Union</name>
  <abbrev>EU</abbrev>
  <member type="member" country="GR F E A D I B L NL DK SF S IRL P GB"/>
  <member type="membership applicant"
      country="AL CZ H SK LV LT PL BG RO EW M CY"/>
  <city> <name>Brussels</name> ... </city>
</organization>

Equivalent:

//organization?[$s := @headq and name?? | abbrev??] {member?? | //city[@id=$s]?? }

---

**XQL: CONCLUSION**

- Ad-hoc-constructs (in different versions)
- insufficient restructuring functionality
  - tree structure of the input is in principle retained
- insufficient join functionality
- no clear semantics for the result format
- queries cannot be nested (cf. SQL, OQL: results are again relations);
  here is even no notion of a subquery
- one of the reasons: no variable concept
- implemented and used up to 2002 in the “Tamino” system of Software AG.
6.2 Query Languages: Requirements

Requirements on XML Query Languages [David Maier and W3C XML Query Requirements]

- closedness: output must be XML
- orthogonality/composability: everywhere where a set of XML elements is required, also a query is allowed.
- clean definition and nesting of operations: selection, extraction/projection, restructuring, combination/join, fusion of elements,
- applicable without presence of schema, but can use a schema,
- retaining the order of nodes,
- [queries should have an XML representation, especially, XML documents should be able to contain embedded queries]
- resolving of XPointer and XLink
- formal semantics: deriving structure of the result, equivalence and query containment

6.3 XML-QL

- simple, pattern-based XML query language:
  WHERE xml-pattern IN url CONSTRUCT result
  
  - usage of variable bindings:
  xml-pattern contains variables that can be used in result,
  
  - declarative,
  
  - “relationally complete”, i.e., joins can be expressed.

Example:

WHERE <country car_code=$id>
  <name>$name</name>
</country>
IN "http://www.../mondial.xml"
CONSTRUCT <country car_code=$id name=$name/>
XML-QL: Joins

Joins are expressed as a list of
WHERE (expr_1 IN doc_1, ..., expr_n IN doc_n)-clauses:

- equijoin inside a document:

WHERE
  <country car_code=$c/></country>
IN mondial.xml,
  <organization abbrev=$org>
    <members type=$type country=$c/>
  </organization>
IN mondial.xml,
CONSTRUCT ...

- Joins that combine multiple documents:

WHERE
  <city name=$name1>
  IN http://www.../europe.xml,
  <city name=$name2>
  IN http://www.../america.xml,
  <connection from=$name1 to=$name2>
  IN http://www.../lufthansa.xml
CONSTRUCT
  <connection>
    <from continent="europe" city="name1"/>
    <to continent="america" city="name2"/>
  </connection>
XML-QL: NESTED QUERIES

WHERE xml-pattern IN url CONSTRUCT result

• result can contain nested WHERE ... IN ... CONSTRUCT statements.

FURTHER FUNCTIONALITY

• tag-variables: WHERE <$tag> ... </>
• regular path expressions: instead of XPath’s “//”, <*> ... </> is used.

DATA MODEL

• Graph-based: XML-tree with IDREF edges:
  WHERE
  <country car_code=$cc>
    <capital><name>$name</name><population>$pop</population></capital>
  </country>
IN ... CONSTRUCT ...

XML-QL: CONCLUSION

• clause-based high-level language
• selection and construction pattern-based (by binding variables in the patterns; similar to Logic Programming)
• join conditions: not in a WHERE clause, but implicitly expressed by join variables (like in Logic Programming)
• graph data model; no difference between tree edges and reference edges
• has been implemented
• used in different projects (e.g. MIX – Mediation in XML; UC San Diego 1999/2000)
  – allows for access and combination of different HTML/XML-sources in a query.
6.4 SQL, OQL etc.

- set-oriented (sets of tuples or objects) language
- implicit iteration over sets:
  
  SELECT ... FROM relation-or-extent c

- variable c ranges over data items
- join: use several such variables and correlate them
- WHERE and SELECT part: use these variables

Similar constructs for XML?

- variables range over sets of nodes
- ... sets of nodes can be addressed by XPath
- straightforward and intuitive:

  for $c in //country
  where $c/population > 1000000
  return $c/name/text()

6.5 Quilt and XQuery

- a Quilt is a “Flickenteppich” ...
- IBM, Software AG, INRIA; literature: WebDB2000-Workshop
- Structure similar to SQL/OQL: clause-based, functional language
  (arbitrary nesting of FLWR expressions allowed),
- Use of variables similar to SQL/OQL,
- based upon XPath (previously XQL/XSL Patterns) in the selection part and upon XML-QL
  (XML patterns) in the construction part:

  • For Let Where Return-clauses
    
    for variable in xpath-expr  // from XQL/XPath and XML-QL
    let additional_variable := xpath-expr
    where condition
    return xml-expr  // from XML-QL

- has been moved into W3C’s “XML Query” in 2001 with only small changes.
- Remark: XQuery is case-sensitive.
  ALL KEYWORDS MUST BE WRITTEN WITH non-capital LETTERS!
XQUERY: EXAMPLE

- for-clause: binding of variables (cf. SQL: FROM)
- where-clause: evaluation of conditions
- return-clause: generation of the result (cf. SQL: SELECT)

```xml
<result>
  { for $c in /mondial/country
    where $c/population > 1000000
    return <bigcountry>
      { $c/name }
      <area> { string($c/@area) } </area>
    </bigcountry>
  } </result>
```

[Filename: XQuery/first-example.xq] generates

```xml
<bigcountry>
  <name>France</name>
  <area>547030</area>
</bigcountry>
<bigcountry>
  <name>Spain</name>
  <area>504750</area>
</bigcountry>
```

ASIDE: TOOLS – XQUERY AS A DATABASE AND WEB QUERY LANGUAGE

XML Databases
- local repository of XML documents
- adding documents to the Database
- access only against locally stored documents
- presence of access paths like indexes etc
- manipulation of documents

Tool: a local eXist repository; see Web page
paths: /mondial//country/name or doc('/db/xmlcourse/mondial.xml')/mondial/country/name

Queries against the Web
- querying the whole Web
- documents not locally stored; only on-the-fly-indexing possible
- access to remote documents by their url

Tool: saxonXQ; see Web page
paths: doc('filename or url')/country/name
**XQuery: Overview of Functionality**

- for-clause: defines nested loops where each of the variables runs over the set of selected values
- variables in XPath expressions: bound in for/let (or by surrounding statements), they are used as starting points for paths and in conditions
- joins:
  - multiple variables in a for-clause:
    
    ```
    for $var_1 in doc_1/path_1, \ldots, \var_2 in doc_n/path_n
    ```
  - correlated definition of the variables in the for-clause:
    
    ```
    for $var_1 in doc_1/path_1, \var_2 in \var_1/path_2, \ldots
    ```
- let-clause for definition of “constants”:
  
  ```
  let $var := expr
  ```
  binds $var to the whole result of expr (in general, a node set).
- nested/iterated for-let-for-let-clauses allowed
- generation of nested and grouped structures:
  the return-clause may contain further FLWR-clauses (which can contain variables from the outer clause).

**Simplest XQuery Queries: XPath**

- Each XPath query is also an XQuery query
- result: a sequence of nodes or literal values
  
  ```
  doc('mondial.xml')//country/name
  ```
  Note: different behavior when returning attribute nodes!
  
  ```
  doc('mondial.xml')//country/@area
  ```

**XQuery: For-clause**

- iterates over the result of xpath-expr
  
  ```
  for $c in /mondial//country/name
  return $c
  ```

[Filename: XQuery/for-example.xq]
XQUERY: RETURN-CLAUSE

Output of all statements must be XML.

- simple case: content of a variable

```xml
for $c in /mondial//country/name
    return $c
```

- and generation of structured results (cf. OQL)

Generation of Structures

- literal XML
- computed element- and attribute constructors (later)

Use of Computed Values/Structures

- enclosed between “{" . . . “}”
- evaluation of variables and XPath expressions
- nested FLWR-clauses

RETURN-CLAUSE: CONSTRUCTION OF RESULT ELEMENTS

- literal XML, values of variables and results of XPath expressions

```xml
<html><table>
<tr><th>Name</th><th>Area</th><th>Population</th></tr>
{ for $c in /mondial/country
    return
    <tr><td>{$c/name/text()}</td>
    <td>{string($c/@area)}</td>
    <td>{$c/population/text()}</td>
    </tr>
} </table></html>
```

[Filename: XQuery/table-example.xq]
returns one table row for each country.
XQUERY: FOR-CLAUSE

Multiple Variables in a For-Clause

• cartesian product
  (cf. FROM-clause in SQL)

```xml
for $c in /mondial//country,
   $o in /mondial//organization
where $c/@capital = $o/@headq
return
  <answer>
    <country>{$c/name/text()}</country><organization>{$o/name/text()}</organization>
  </answer>
```

[Filename: XQuery/cartesian-example.xq]

• compare where clause with equivalent
  where $c/id(@capital) is $o/id(@headq)
on node level ("=" would also be correct here, taking the string value of the nodes).

XQUERY: FOR-CLAUSE

Multiple Variables in a For-Clause

• "correlated" Join
  (cf. FROM-clause in Schema-SQL and OQL)

• subset of the cartesian product

```xml
for $c in /mondial/country,
   $p in $c/province
return
  <answer>
    <country>{$c/name/text()}</country><prov>{$p/name/text()}</prov>
  </answer>
```

[Filename: XQuery/correlated-join-example.xq]
**RETURN-CLAUSE WITH NESTED FLWR-CLAUSE**

- inner query used in the outer return-clause (cf. OQL)

  ```xquery
  for $c in /mondial/country
  where $c/province
  return
    <answer>
      {$c/name}
      {
          for $p in $c/province
          return
            <prov>{$p/name/text()}</prov>
        }
    </answer>
  ```

  [Filename: XQuery/nested-flwr-example.xq]

  generates for each country that has provinces an `<answer>` element that contains a `<name>` element and a sequence of `<prov>` elements.

**LET-CLAUSE**

- does not iterate over the result of `xpath-expr`
- but binds the complete result of `xpath-expr` as sequence of nodes to the variable:

  ```xquery
  let $var := xpath-expr

  for $c in /mondial/country
  let $cities := $c//city/name
  return
    <country>
      {$c/name}
      {$cities}
    </country>
  ```

  [Filename: XQuery/let-example.xq]

  • useful for keeping intermediate results for reuse (often missed in SQL)
WHERE-CLAUSE: CONDITIONS

Similar to XPath's conditions (same predicates etc):

- logical “and” and “or”
- “not(...)” as a boolean function
- Comparisons: “is” for node identity, “<” and “>” for document order, “follows” and “precedes”
- Quantifiers: where some|every $var in expr satisfies condition

```xquery
for $c in /mondial/country
where some $city in $c//city satisfies $city/population > 1000000
return $c/name

for $c in /mondial/country
where every $city in $c//city satisfies $city/population > 1000000
return $c/name
```

[Filenames: XQuery/some-example.xq and every-example.xq]

USE CASE: JOIN BETWEEN DIFFERENT DOCUMENTS

- doc(...) function to access files (local or from the Web)
- here: join by a subquery

```xquery
<result>
{ for $c in doc(concat('/http://www.dbis.informatik.uni-goettingen.de','/Mondial/mondial-europe.xml'))/mondial/country
  where some $l in doc('hamlet.xml')//LINE
  satisfies contains($l, $c/name)
  return
  <country>
  {$c/name}
  </country>
}
</result>
```

[Filename: XQuery/join-web-documents.xq]
CONDITIONAL EVALUATION AND ALTERNATIVES

- if-then: alternative choice of subelements
  
  \[
  \text{if (expr) then expr else expr}
  \]

  \[
  \begin{align*}
  \text{<result>}
  & \{ \text{for } c \text{ in /mondial/country} \\
  & \quad \text{return}
  & \quad \text{<country>}
  & \quad \quad \{c/name\}
  & \quad \quad \{\text{if } (c/province) \text{ then } c/province/city \text{ else } c/city\}
  & \quad \text{</country>}
  & \}\n  \text{</result>}
  \]

  [Filename: XQuery/if-else-example.xq]

ORDER OF RESULT SET

XPath: the result is always returned in document order:

- purely navigational access:
  
  \[
  \text{//country/city/name}
  \]

- even when a backward axis is used during navigation, the nodes are enumerated in document order:
  
  \[
  \text{//country[name='Germany']/province[last()]/preceding-sibling::*/name}
  \]
  
  (backward axis is only relevant for context functions in immediate conditions)

- or when id-referencing is used:
  
  \[
  \text{id(//organization/@headq)/name}
  \]
  
  (note: cities are not ordered according to the order of the organizations!)

XQuery: result set is ordered according to for-clause:

\[
\text{for } c \text{ in } //\text{organization}
\quad \text{return } \text{id}(c/@headq)/name
\]

let-clause: binds the result set according to the respective order.
SORTING

• order by:  \textit{expr} order by (\textit{expr} [\textit{ascending}|\textit{descending}])

\begin{verbatim}
<result>
  { for $c in //country
    order by $c/name
    return $c/name }
</result>
\end{verbatim}

[Filename: XQuery/orderby-example.xq]

• note that the interpreter must be told whether the values should be regarded as numbers or as strings (default: alphanumerical)

\begin{verbatim}
<result>
  { for $c in //country
    where $c/population > 0
    order by number($c/population)
    return $c/name }
</result>
\end{verbatim}

[Filename: XQuery/orderby-num-example.xq]

GROUPING AND AGGREGATION

• aggregate functions over result sets (avg, sum)

• bind variable with “for”-clause

• assign group with “let” (dependent on the current value in the for-clause) to a variable

• apply aggregate function to a nodeset

\begin{verbatim}
<result>
  { for $c in /mondial/country
    let $cities := $c//city
    where sum($cities/population) > 10000000
    return
      <answer>
        {$c/name}
        {sum($cities/population)}
      </answer>
  }
</result>
\end{verbatim}

[Filename: XQuery/aggr-1-example.xq]
AGGREGATION

- aggregation over result of a FLWR subquery
- bind (single) intermediate result by “let”

```xml
<result>
{ for $c in /mondial/country
  let $maxpop := max( for $citypop in $c//city/population/text()
    return $citypop )
  return
    <answer>
      {$c/name}{$maxpop}
    </answer>
}
</result>

[Filename: XQuery/aggr-2-example.xq]
```

ATTRIBUTES IN THE RETURN-CLAUSE

- note that expressions the form “@bla” return attribute nodes - these are (AttrName,value)-pairs:

```xml
<result>
  {//country[name='Germany']/@car_code}
</result>
```

generates `<result car_code="D"/>`.  
- attribute nodes are always added to the surrounding element.  
- if only their value is needed, apply string().

```xml
for $c in /mondial/country
return
  <country>
    {$c/@area}
    {string($c/@car_code)}
  </country>

[Filename: XQuery/attribute-example.xq]
```

Result:
```
<country area="28750">AL</country>
<country area="131940">GR</country>
```
**COMPUTED ELEMENT- AND ATTRIBUTE NAMES**

- explicit constructors
  - element *expr attrs-and-content*
    - the evaluation of *expr* yields the name of the element, the result of *attrs-and-content* is then inserted as attributes and content
    - Note: content is a node sequence, separated by “,”
  - attribute *expr expr-value*
    - the evaluation of *expr* yields the name of the attribute, *expr-value* yields its value.

```xquery
{ for $c in doc('mondial.xml')//country
  where $c/encompassed
  return
  element { $c/@car_code }
    { attribute {$c/encompassed[1]/@continent} "yes",
      $c/name
    }
} </result>
```

A result node:

```
<result>
{ for $c in doc('mondial.xml')//country
  where $c/encompassed
  return
  element { $c/@car_code }
    { attribute {$c/encompassed[1]/@continent} "yes",
      $c/name
    }
} </result>
```

[Filename: XQuery/computed-constructors-example.xq]

**COMPUTED ELEMENT- AND ATTRIBUTE NAMES: ANOTHER EXAMPLE**

- the element content can be computed by an XQuery expression (cf. usage of *expr* on the previous slide):

```xquery
{ for $c in doc('mondial.xml')//country
  where $c/encompassed
  return
  element { $c/@car_code }
    { for $e in $c/encompassed
      return attribute {string($e/@continent)} "yes",
      $c/name
    }
} </result>
```

[Filename: XQuery/computed-constructors-example2.xq]
HANDLING DUPLICATES

- recall from XPath: results (and intermediate results) of XPath expressions are node sets in document order
  \[\text{for } \$x \text{ in } \text{xpath-expr}, \text{ let } \$y := \text{xpath-expr} \\]
  always results in a set (i.e., duplicates removed)

- recall Slide 219 for removal of duplicate values: distinct-values(...)
  \[
  \text{distinct-values(doc('...')/SPEAKER)}
  \]

How many speeches has each of the speakers in “Hamlet”?

\[
\text{for } \$a \text{ in distinct-values(doc('/db/xmlcourse/hamlet.xml')/SPEAKER)}
\text{let } \$n := \text{count()//SPEECH[SPEAKER = } \$a \text{]})
\text{order by } \$n \text{ descending}
\text{return}
\text{<answer>}
\{\$a\}
\{\$n\}
\text{</answer>}
\]

- takes only the string values (⇒ no further navigation applicable)

Handling Duplicates in XQuery (cont’d)

- FLWR expressions (e.g., for $c$ in ... return $c$) do not eliminate duplicates automatically
  \[
  \text{for } \$o \text{ in } //\text{organization return } \$o/\text{id(@headq)}
  \text{returns duplicates}
  \]

- distinct-values(for $o$ in //organization return $o/\text{id(@headq))}
  returns only the string values

- so it must be done programmatically (often, specific for the given problem: iterate over the target set and do the test in a subquery) – cf. SQL:
  \[
  \text{select * from <table-of-entity-tuples> where <condition>}
  \]

- or by a generic function – see Slide 288
The datatypes specified by XML Schema are used in XPath/XQuery (and XSLT)

- Syntax: constructors like `xs:dateTime('syntactical representation')`
- syntactical representations:
  - `xs:dateTime`: `yyyy-mm-dd Thh:mm:ss[.xx][+[−]hh:mm]`
  - `xs:date`: `yyyy-mm-dd` and `xs:time: hh:mm:ss[+[−]hh:mm]`
  - `xs:duration`: `P [nY][nM][nD][T[nH][nM][nS]]`, where `n` can be any natural number
  - `xs:dayTimeDuration`, `xs:yearMonthDuration`: restrictions of `xs:duration`

```xml
let $x := xs:dateTime('2009-08-01T13:51:20.99'),
 $y := xs:date('2008-12-31'),
$t1 := xs:time('12:50:00+01:00'), (: timezone +1 = Frankfurt :) 
$t2 := xs:time('15:35:00.50-05:00') (: timezone -5 = New York :)
return <e const="{$x}" diff="{$t2 - $t1}" d="{$y + xs:yearMonthDuration("P1Y2M")}" 
sum1="{$x:time("11:12:00") + xs:dayTimeDuration("PT1H75M")}" 
sum2="{xs:dateTime("2009-01-10T11:12:00") + xs:dayTimeDuration("P3DT26H40M")}/> 
```

[Filename: XQuery/datetime-test.xq]

- resulting diff = “PT8H45M0.5S” (an `xs:duration`), sum1 = “13:27:00” (an `xs:time`),
  sum2 = “2009-01-14T13:52:00” (an `xs:dateTime`), d = “2010-02-28” (an `xs:date`)

---

**Actual Usage**

... is often simple

```xml
<country car_code="B">
    <indep_date>1830-10-04</indep_date>
</country>
```

[Filename: XQuery/simple-date-example.xq]

note: explicit `[indep_date < xs:date('1900-01-01')]` would be safer.
FUNCTIONS AND OPERATORS

XPath and XQuery Standard Operators

- Recall Slide 213 for string() and name(), and Slide 210 for id().
- See "W3C XML Query Functions and Operators" for predefined functions, especially concerning dates+times.

Further Operators: EXSLT

Some mathematical functions (sqrt etc.) are not supported as builtins in XPath/XQuery, but only as extensions in EXSLT:

- http://www.exslt.org/, supported e.g. by saxon.
- declare namespace math="http://exslt.org/math"
- use math:sqrt, math:sin, ...
- "XPath and XQuery Functions and Operators 3.0" will support math functions (WD 2010, not yet finalized [2011])

USER-DEFINED FUNCTIONS

- User defined functions are declared in the prolog:
  declare function func_name ([$var1, ..., $varn]) [as returnType]
  {
    expr that uses $var1, ..., $varn
  }
- Parameters: $var, [as paramType], default for parameter and return types is item()*(i.e. a sequence of nodes, literals etc.),
- Any sequence type may be used for paramType and returnType (cf. XML Schema),
- Any XQuery expression is allowed in the function body.
USER-DEFINED FUNCTIONS: EXAMPLES

• A function computing the population density for a given country:

```xquery
declare function local:density ($name as xs:string) as element(density) {
    for $c in doc('mondial.xml')/country[name=$name]
    let $density := if ($c/@area > 0) then $c/population div $c/@area else 0
    return <density>{$density}</density>
};
local:density('Germany')
```

[Filename: XQuery/function-density.xq]

• Example for a recursive function:

```xquery
declare function local:depth($e as node()) as xs:integer {
    if (fn:empty($e/*)) then 1 else fn:max(for $c in $e/* return local:depth($c)) + 1
};
local:depth(/
```

[Filename: XQuery/function-depth.xq]

Ignoring the FLWR, XQuery can even be used as a common functional language:

• every (arithmetic + if) expression is a valid XQuery expression

```xquery
(:call saxonXQ faculty.xq x=5
   or saxonXQ ?x=5 faculty.xq :)
declare variable $x external;
declare function local:faculty($n as xs:integer) as xs:integer {
    if ($n=1) then 1
    else $n * local:faculty($n - 1)
};
local:faculty($x)
```

[Filename: XQuery/faculty.xq]

• for strings as parameters use

```xquery
saxonXQ ?foo='xs:string("blabla")'
```
USER-DEFINED FUNCTION: EXAMPLE

Remove duplicates from a node set (taken from the example Section from W3C XPath/XQuery Functions and Operators):

```xml
declare function distinct-nodes-stable ($arg as node()*) as node()*
{
    for $a at $apos in $arg
       let $before_a := fn:subsequence($arg, 1, $apos - 1)
       where every $ba in $before_a satisfies not($ba is $a)
       return $a
}
```

PRACTICAL HINTS: OUTPUT

When creating output, most XQuery engines generate the XML declaration, and output “<” and “>” as “&lt;” and “&gt;”, respectively.

**Add Doctype Declaration to the Output**

- XQuery engines output only the XML structure itself
- how to add the <!DOCTYPE mondial SYSTEM “mondial.dtd”> preamble?
  With saxon, use
  ```xml
  declare namespace saxon="http://saxon.sf.net/";
  declare option saxon:output "indent=yes";
  declare option saxon:output "doctype-system=mondial.dtd";
  ```

**Generating non-XML code, text, etc**

- In case it is intended to generate e.g. LaTeX, N3 or whatever output, the XML declaration and the <->-conversion must be avoided.
  With saxon, use
  ```xml
  declare option saxon:output "method=text";
  ```
OPERATING WITH SEQUENCES

Comparisons are instance-based: if one operand is a sequence, each value is compared:

- ... as we have seen for XPath: `country[//city/name = "Cordoba"]/name`  
  `country[//city/population > 1000000]/name`

- but somewhat surprising when using a “let”-view:
  
  ```
  let $europnames := //country[encompassed/@continent="europe"]/name
  for $country in //country where not ($country/name = $europnames)
  return $country/name
  ```

  [Filename: XQuery/seq-comparison-example.xq]

  outputs all names of non-european countries.

- selection from let-sequences is also instance-based:
  
  ```
  let $europcountries := //country[encompassed/@continent="europe"]
  return $europcountries[@area>300000]/name
  ```

  [Filename: XQuery/seq-selection-example.xq]

OPERATIONS ON NODES AND NODE SEQUENCES

- “=” compares the string-values of nodes, not “correct” if node identity has to be checked

- “is” compares node identity:

  ```
  for $c in //country, $o in //organization
  where $c/id(@capital) is $o/id(@headq)
  return <pair country='\{$c/@car_code\}' org='\{$o/abbrev\}'/>
  ```

  [Filename: XQuery/node-comparison.xq]

- “is” is not allowed for sequences:

  ```
  let $caps := //country/id(@capital)
  for $hq in //organization/id(@headq)
  where $hq is $caps  (: not allowed :)  
  return $hq/name
  ```

  [Filename: XQuery/nodes-comparison-example-1.xq]

  ⇒ “A sequence of more than one item is not allowed as the second operand of “is” ”
OPERATIONS ON NODES AND NODE SEQUENCES

- explicit iteration via some:

```xquery
let $caps := //country/id(@capital)
for $hq in //organization/id(@headq)
where some $cap in $caps satisfies $hq is $cap
return $hq/name
```

[Filename: XQuery/nodes-comparison-example-2.xq]

- index-of(sequence, item) → integer

```xquery
let $caps := //country/id(@capital)
for $hq in //organization/id(@headq)
where index-of($caps,$hq) (: checks if $caps contains $hq :) 
return $hq/name
```

[Filename: XQuery/nodes-comparison-example-3.xq]

- ... or “where $caps intersect $hq”, or even shorter:

```xquery
let $caps := //country/id(@capital)
let $hqs := //organization/id(@headq)
return ($hqs intersect $caps)/name
```

[Filename: XQuery/nodes-comparison-example-4.xq]

FLEXIBILITY

For each task, there is a multitude of possible solutions ...

Example: Uncorrelated Subqueries

Names of all countries that are larger than Germany:

- XPath:

```xml
//country[@area > number(//country[@code='D']/@area)]/name
```

- XQuery and SQL: uncorrelated subquery/semijoin

```xquery
for $c in //country
where $c/@area > number(//country[@code='D']/@area)
return $c/name
```

```sql
SELECT c.name FROM country c
WHERE c.area > (SELECT c2.area FROM country c2
WHERE c2.code = 'D')
```

- binding the uncorrelated subquery to a variable:

```xquery
let $germanyarea := number(//country[@code='D']/@area)
for $c in //country
where $c/@area > $germanyarea
return $c/name
```
EXERCISES

... see Web.

Exercise 6.1
Determine the lowest mountain that is the highest mountain of the continent where it is located.

Solve the problem for the relational Mondial-DB in SQL, and for XML in XQuery.

XQUERY: CONCLUSION

Design and Functionality
• combines the positive experiences of previous approaches
• avoids their drawbacks
• intuitively clear syntax and semantics
• declarative, orthogonal, functional style: every expression is a function on nodesets that also returns a nodeset
  – explicit, variable-based iteration: “for var in expression”
  – implicit iteration: “collection[condition]” or “collection/path”
• Theoretical background (see W3C XML Query Formal Semantics; datatypes of the XML Schema and XML Query Data Model)
  – for each expression (and thus also for its result), the formal type (according to the XML Schema datatypes) can be determined.
  – the type of each variable is determined in the same way.
  – formal, denotational semantics of queries: “what is the answer set of a given expression?”
W3C XML Query Formal Semantics:

• XPath/XQuery is a functional language.

• is built from expressions, rather than statements. Every construct in the language (except for the XQuery query prolog) is an expression and expressions can be composed arbitrarily.

• The result of one expression can be used as the input to any other expression, as long as the type of the result of the former expression is compatible with the input type of the latter expression with which it is composed.

• Another characteristic of a functional language is that variables are always passed by value, and a variable’s value cannot be modified through side effects.

Restrictions

• up to now no resolving of XLink/XPointer (see later)

• only a query language:

decision of the W3C: first complete XQuery 1.0 as a query language and make it consistent with XML Schema and XML Query Data Model as a “Recommendation”, and then start official thoughts about updates in XQuery 2.0.
GENERAL DESIGN PATTERNS FOR DATABASE QUERY LANGUAGES

SQL, OQL, XML-QL, XQuery (and many others) use the same underlying principle:

- binding variables
- evaluating a condition
- generating a result (which is a set of data items of the underlying data model)

Note: XQL did not follow this idea ⇒ restricted expressiveness and clarity

... let’s now have a look on one more XML query language

- the underlying principle is the same
- everything else is “just syntax”!

6.6 Further (Academic) Query Languages

XPATHLOG

- Prolog-/Datalog-style (May, DBPL and VLDB 2001; TPLP 2004)
- based on F-Logic
  - path syntax changed from \texttt{step.step.step} to \texttt{step/step/step}
  - same syntax for conditions as for F-Logic: “[…]” could be reused
  - F-Logic semantics (1989) closely related with XPath semantics
  - new: distinction between attributes/subelements
- Binding of variables at \textit{arbitrary} positions of an expression
- joins as conjunction (as in Prolog/Datalog)
**XPATHLOG**

- implicit resolving of multi-valued attributes
- implicit resolving of reference attributes

```xml
?- //country->C[name->N and @membership->0/name->A].
```

- access to signature/metadata

```xml
?- //country[name="Germany"]/M.
?- //country[name="Germany"]/O.A.
```

- class membership and -hierarchy

```xml
?- C isa country[name->N]/M.
?- _C isa country/@A->_O, _O isa X.
?- country[@M=>C]. % from DTD
```

---

**XPATHLOG: OVERVIEW**

- declarative language
- implicit iteration (fixpoint semantics)
- (equi-)join variables
- XPath-style semantics in rule heads for *generation* and *manipulation* of XML data
- first implementation of an update language for XML (Demo VLDB 2001)
  generation of XML in rule heads:
  ```
  C[density -> D] :- C isa country[population -> P; @area -> A], D is P div A.
  ```
- fixpoint semantics for execution
- can compute transitive closure etc.

```sql
R[tr_flows_into -> S] :- R isa river, R/to[@watertype -> "seas"; @water -> S].
R[tr_flows_into -> S] :- R isa river, R/to[@watertype -> "river"; water -> R2], R2[tr_flows_into -> S].
```
GENERAL DESIGN PRINCIPLES FOR DATABASE QUERY LANGUAGES

SQL, OQL, XML-QL, XQuery (and many others) use the same underlying principle:

- binding variables
- evaluating a condition
- generating a result (which is a set of data items of the underlying data model)

<table>
<thead>
<tr>
<th>SQL/OQL</th>
<th>XML-QL</th>
<th>XQuery</th>
<th>XPathLog</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>SQL: flat data</td>
<td></td>
<td>XPath navig.+</td>
</tr>
<tr>
<td></td>
<td>model</td>
<td></td>
<td>XPath patterns</td>
</tr>
<tr>
<td></td>
<td>OQL: + path navi g.</td>
<td></td>
<td></td>
</tr>
<tr>
<td>conditions:</td>
<td>WHERE clause</td>
<td>Patterns</td>
<td>XPath filters</td>
</tr>
<tr>
<td></td>
<td>(equality join</td>
<td>conditions)</td>
<td>(join conditions)</td>
</tr>
<tr>
<td></td>
<td>errods)</td>
<td>(only non-join-conds)</td>
<td></td>
</tr>
<tr>
<td></td>
<td>WHERE clause</td>
<td>WHERE clause</td>
<td>separate conjuncts</td>
</tr>
<tr>
<td></td>
<td>(comparisons+j oins)</td>
<td>(all)</td>
<td>(comparisons+j oins)</td>
</tr>
</tbody>
</table>

- the underlying Logic Programming fixpoint semantics enables XPathLog to compute the transitive closure
- ... but it does not allow for syntactically nested statements

FURTHER (ACADEMIC) QUERY LANGUAGES

- XML-GL (Comai, Politecnico Milano, 1999): graphical “language”
- Lorel-XML (Stanford Univ., 1999): OQL-style language, migration of Lorel
- YATL-XML (Cluet, INRIA, 2000): term-based language, migration of YATL
- Lixto/Elog (Gottlob, TU Wien, 2001): graphical tool for data extraction from the Web, Datalog-based internals
- Xcerpt, XChange (Bry et al, LMU München, 2002): term- and unification-based language

... many different approaches to the same goal (mainly in Europe).
Overview in (May, TPLP 2004).
Chapter 7
Manipulating XML Data

• XML data in files:
  – usually no changes (except manually or by scripts)
  – transformations XML → HTML etc: XSLT

• XML data in application systems
  – inside the application programming language; mostly by the DOM-API
  – no special data manipulation language necessary (cf. OQL)?

• different proposals
  – pre-XQuery commercial area:
    • XMLDB: XUpdate (1999)
    • eXcelon (2000; XUL as extension of XSLT)
  – academic area:
    • “Updating XML” (Halevy et al, SIGMOD 2001) as an extension to XQuery
    • XPathLog (May, VLDB 2001): Prolog-style query- and manipulation language

EXTENDING XQUERY WITH UPDATES – CONCEPTS

In the meantime consensus about which operations is reached. Syntax is still open.

• always wrt. a context node

• base operations:
  – delete \texttt{node}
  – rename \texttt{node} as \texttt{name}
  – insert \texttt{node/nodes} before|after|into \texttt{node}

• combined operations:
  – replace \texttt{node} with \texttt{node}
  – move \texttt{node} before|after|into \texttt{node}
7.1 XML:DB Initiative’s XUpdate

- XML:DB Initiative founded in late 1999
  Goal: interface for storing XML in databases
- Low-level API (Java etc., using DOM + XPath ...)
- an update concept: XUpdate
- Implementation:
  dbXML Core XML Database released as Open Source software in Sept. 2000
  transferred to the Apache Software Foundation (“Xindice”)
- http://xmldb-org.sourceforge.net/
  (inactive?)
- The XML:DB database API is implemented in several systems:
  eXist, X-Hive, Tamino, XML:DB Lexus, ...

... but here we are mainly interested in XUpdate ...
(note that XUpdate (1999) is not related with XQuery (2001))

306

XML:DB XUPDATE

Situation in 1999: XML, XPath, XSLT [see later], low-level APIs

- Requirement: “The XML Update specification MUST be an XML element”
  i.e., the language is itself in XML syntax (like XSLT and XML Schema)
- XUpdate: a very basic description of update operations:
  – which node (elements, attributes)
  – which operation (delete, update value, append/insert to contents)
  – new value (in case of update/append/insert)

Basic structure:

```
<xu:modifications xmlns:xu= "http://www.xmldb.org/xupdate">
  <xu:operation select= "xpath-expression">
    contents (e.g. new value)
  </xu:operation>
</xu:modifications>
```

... submit such an element as a kind of a “message” to the DB and get the update.

307
XUpdate: Example

```xml
<?xml version="1.0"?>
<xu:modifications version="1.0" xmlns:xu="http://www.xmldb.org/xupdate">
  <xu:append select="/mondial/country[name='Germany']">
    <xu:element name='localname'>Deutschland</xu:element>
  </xu:append>
</xu:modifications>
```

[Filename: XUpdate/append.xu]

Calling eXist with (see client.sh -h)

```
/bin/gen_client.sh -u user -P password -c /db/may -f mondial.xml -X append.xu
```

executes the update.

- `select= "xpath"` is the same as in XSLT (see later), XML Schema etc. – a widely used concept in the XML world.
  (if multiple nodes are addressed, each one is modified)

- `<xu:element>` constructor is the same as in XSLT (1998) and later in XQuery’s RETURN clause

- analogously insert-before and insert-after.

XUpdate: Examples (Cont’d)

```xml
<?xml version="1.0"?>
<xu:modifications version="1.0" xmlns:xu="http://www.xmldb.org/xupdate">
  <xu:remove select="/mondial/country[name='Germany']/localname"/>
</xu:modifications>
```

[Filename: XUpdate/remove.xu]

```xml
<?xml version="1.0"?>
<xu:modifications version="1.0" xmlns:xu="http://www.xmldb.org/xupdate">
  <xu:update select="/mondial/country[name='Germany']/population">
    80000000
  </xu:update>
</xu:modifications>
```

[Filename: XUpdate/update.xu]
XUpdate: Examples (Cont’d)

• get the new value from the database:

```xml
<?xml version="1.0"?>
<xu:modifications version="1.0" xmlns:xu="http://www.xmldb.org/xupdate">
  <xu:update select="/mondial/country[name='Germany']/population/text()">
    <xu:value-of select="/mondial/country[name='Germany']/@area"/>
  </xu:update>
</xu:modifications>
```

[Filename: XUpdate/update-select.xu]

note: the inner select cannot depend on the current node.

```xml
<?xml version="1.0"?>
<xu:modifications version="1.0" xmlns:xu="http://www.xmldb.org/xupdate">
  <xu:variable name="bla" select="/mondial/country[name='Germany']/gdp_total/text()"/>
  <xu:update select="/mondial/country[name='Germany']/population/text()">
    <xu:value-of select="$bla"/>
  </xu:update>
</xu:modifications>
```

[Filename: XUpdate/update-variable.xu]

---

**XUpdate: Conclusion and Comments**

• XML-syntax of the language strongly influenced by XSLT (1998)
  – elements as commands
  – select="..." selects nodes to which the command is applied
  – use of variables select="$variable" as in XSLT, and later also in XQuery
  – element/command contents specifies what is to be done
  – element generation by literal XML (also in XSLT and later XQuery)

• only very simple functionality
  – no way to compute the inner value,
  – no iteration etc.

• same time: combination with XSLT and XUpdate to XUL (XML Update Language/Updategrams [Excelon]):
  XSLT program structures + XUpdate operations, applied to “current node” of XSLT.
7.2 XQuery with Updates 2001

- extend a *declarative query language* with updates
- based on *variable bindings*
- SQL: FROM-WHERE for selecting nodes ...  
  ... that are then modified.
- XQuery: FOR-LET-WHERE for selecting nodes ...  
  ... that are then modified.
- update instead of the return-clause (cf. SQL: UPDATE vs. SELECT)?
- or what?

XQUERY WITH UPDATES – AN EARLY PROPOSAL

- QuiP: the 2001/02 XQuery prototype of Software AG [Diplomarbeit P. Lehti 2001],  
  later integrated into the Tamino system (before: XQL).
- calling quip filename.xq > bla.xml wrote the modified XML to a file.

```xml
update
for $c in document("twocountries.xml")//country
let $area := string($c/@area)
delete $c/@area
insert <area>{$area}</area> after $c/name
rename $c//city[@id=$c/@capital] as capital
replace $c/@car_code with
    attribute code {concat($c/name/text(), ":", string($c/@car_code))}
replace $c/population/text() with
    $c/population/text() * (1 + $c/population_growth div 100)
insert "biggest city" into
    $c//city[population = max(for $citypop in $c//city/population/text()
             return int($citypop))]
[Filename: XQuery/update.quip]
```
XQuery with Updates – W3C Proposal

• XQuery reached recommendation state in 2007 ... as a query language still without updates.

• “XQuery Update Facility”, first W3C Working Draft has been published 27 January 2006; http://www.w3.org/TR/xquupdate

New Expressions

do insert SourceExpr ( ([as (first | last)] into) | after | before) TargetExpr

do delete TargetExpr

do rename TargetExpr as Expr // Expr must result in a qname

do replace [value of] TargetExpr with Expr

• Syntax still changing,

• not implemented in saxonA 9.1, only in (commercial) saxonB 9.1

XQuery with Updates – Transformation Command

• not an update!

1. assign variable(s),

2. update things bound to the variable(s),

3. return something generated from the (updated) variables.

transform copy $VarName := Expr (, $ VarName := ExprSingle)*

modify UpdateExpr return Expr